UNIVERSITY OF ILLINOIS

DIGITAL COMPUTER LABORATORY

ILLIAC II LIBRARY ROUTINE P1-TP-65v

TITLE: TWELVE PACK

TYPE: Random number, random address drum test

NUMBER OF WORDS: 123 including typewriter output routine

DURATION: Endless

PARAMETERS: Greenswitch up, no error print
Greenswitch down, error print

USE: Read in by NICAP

DISCUSSION: Twelve pack was written to test the drum under the condition of a continuous stream of either reads or writes. Execution begins with the generation of 256 random numbers starting at location 512. Then a table of twelve random drum addresses is stored one per word starting at location FRA with the address in the second quarter word position. Stepping sequentially through this table, the same random numbers are written on the twelve specified blocks on the drum. While the last block is being written, the table of addresses is shuffled and placed in a new table beginning at FRR. Finally, using the newly arranged table, the twelve blocks are read back into blocks 768, 1024, ... Comparison of the retrieved data is made with the master copy. (Note: There may be a duplication of drum addresses in the tables.)

The error printout for Twelve Pack is as follows:

00055	A	B	C	ZERO
00056	W_1	W_2	W_3	W_4
00057	R_1	R_2	R_3	R_4

EHRMAN #
**

where

A = Drum Address

B = Word in block which failed

C = Position in string of 12 readings

Wi = 1/4 words of word written

Ri = 1/4 words of word read

As usual, the greenswitch down means print if an error occurs. Greenswitch up means finish printing the present block of three words, then resume the program and don't print until the switch is again put down.

DATA RATE: The number of drum sector times taken by various parts of the program are approximately:

Generate 256 random numbers	3.5	sector times
Write 12 blocks 1 * 4 + 11 * 4-1/2 =	53.5	overlapped
Shuffle address table	1.	overlapped
Read 12 blocks	53.5	overlap 4.5
Compare 12 blocks 5-1/2 x 12	66.	overlap 4.5
	172	sector times = .380 seconds

The data rate for blocks written, read, and checked is

$$31.6 \frac{\text{blocks}}{\text{second}} = 1900 \frac{\text{blocks}}{\text{minute}}$$

The maximum theoretical rate for blocks written and read is

$$4 \frac{\text{blocks}}{\text{revolution}} \times 3400 \frac{\text{revolutions}}{\text{minute}} = 13{,}600 \frac{\text{blocks}}{\text{minute}}$$

Thus, this program keeps the drum busy

$$\frac{1{,}900}{13{,}600} = 14.0 \text{ per cent of the time.}$$

For comparison, on July 2, it was determined by analog techniques that while running this program a transfer is in progress 15.7 per cent of the time.

DATE:	July 6, 1964
PROGRAMMED BY:	G. E. Cooper
APPROVED BY:	H. C. Brearley


Gen. 256 Rand. No's. in 512-767
i = 0
M5 = Rand. Drum Address
F5 → FRA + i
i = i + 1
i = 12
NO
YES
j = 512
i = 0
F5 = FRA + i
POD M5
IBT j
i = i + 1
i = 12?
NO
YES
Randomly Shuffle FRA---FRA+11 into FRR---FRR+11
j = j + 256
i = 0
F5 = FRR + i
PID M5
IBT j
i = i + 1
j = j + 256
i = 12
NO
YES
Compare 512-767 with 768-1023,...
Down
Center
Green Sw.

```
                DRUM TEST    12 PACK      G. COOPER   APRIL  1964
                   GENERATE 256 RANDOM NUMBERS, WRITE IN 12 RANDOM BLOCKS,
                   READ(IN DIFFERENT RANDOM ORDER), CHECK
                OUTPUT FORMAT ON CONSOLE TYPEWRITER (OCTAL)
                              *** DRUM BLOCK (0-377)
                             *
                            *      **** WORD POSITION IN BLOCK (0-377)
                           *      *
                          *      *      **** POSITION OF FAILED BLOCK IN
                         *      *      *              READ STRING (1-140CTAL)
                  00055 00377 00377 00014 00000
                  00056 XXXXX XXXXX XXXXX XXXXX    WORD WRITTEN
                  00057 XXXXX XXXXX XXXXX XXXXX    WORD READ
          ORG     0
BEG       LFR     4,FSW
          LFR     3,FSW+1           LOAD OPERANDS
          LFR     2,FSW+2
BEG1      CSM     2,256
FRN1      CAD     F2                GENERATE 256 RANDOM NUMBERS
          MPY     F3
          SAL     F3
          LRS     2
          SAL     F7
          CRM     12,3
          CAD     F7
          ADM     1,M0
          CAM     0,M1-M0
          CRM     1,12
          CAE     M1
          SAM     OUT+M2+256
          CJU     2,FRN1
          CSM     2,12              GENERATE 12 RANDOM ADDRESSES
FRAC      ADM     1,M0
          CAM     0,M1-M0
          CRM     1,12
```

```
        SFR     4,FRA+M2+12     STORE RANDOM ADDRESSES ONE PER
        CJU     2,FRAC          WORD IN FRA TO FRA+11 IN 2ND
        CSM     2,12            1/4 WORD POSITION
FWN     LFR     5,FRA+M2+12
        ANN     5,255
        POD                     WRITE FROM 512 INTO 12 RANDOM
        IBT     OUT             DRUM POSITIONS
        CJU     2,FWN
        LFR     5,ZER
        CSM     2,12
FNG     ADM     1,M0
        CAM     0,M1-M0
        CRM     1,12
FOG     ANN     1,15            RANDOMLY REARRANGE THE
        LFR     6,FRA
        JUM     10,FAG          SET OF 12 ADDRESSES INTO
        ADM     1,1             FRR TO FRR+11
        TRA     FOG
FAG     SFR     6,FRR+M2+12
        ANN     1,15
        SFR     5,FRA
        CJU     2,FNG
        CSM     2,12
        CAN     3
FRL     LFR     5,FRR+M2+12     READ INTO 768,1024,---FROM 12
        ANN     5,255
        PID
        IBT     OUT+256+M3      ADDRESSER IN REARRANGED TABLE
        ADM     3,256
        CJU     2,FRL
        CSM     2,12
        CAN     3,256
FCP1    CSM     4,256
FCP     LFR     6,OUT+256+M4    COMPARE THE RESULTS OF THE 12
        JDC     12,BEG1         READS WITH THE ORIGINAL BLOCK
```

```
        LFR     7,OUT+256+M3+M4     OF RANDOM NUMBERS
        EON     8,M12
        EON     9,M13
        ORM     8
        EON     10,M14
        ORM     8
        EON     11,M15
        ORM     8
        JUM     8,GPR               IF AN ERROR -- PRINT
FPOO    CJU     4,FCP
        ADM     3,256
        CJU     2,FCP1
        TRA     BEG1                RETURN TO BEGINNING
GPR     SFR     5,HOLD+3
        LFR     6,OUT+256+M4        STORE PRINT INFORMATION
        SFR     6,HOLD+1
        SFR     7,HOLD+2
        LDM     5,FRR+M2+L2
        CAM     6,M4
        CAM     4,M5
        CAM     5,M6
        CAM     6,M2+L2
        ANM     4,255
        ADM     5,256
        CAM     7
        SFR     5,HOLD
        CAM     13,HOLD
        CAM     14,HOLD+2
        JSB     15,ESTFW            TO PRINT SUBROUTINE
        FIL
        ATN     768
        SSR     62
        LFR     5,HOLD+3
        TRA     FPOO                RETURN TO COMPARISON
HOLD    BSS     4
```

```
FRA     BSS     16
FRR     BSS     12
ZER     OCTQ    ,,,                     OPERATION CONSTANT
OUT     EQUS    512
FSW     OCTQ    ,1,,
        OCTQ    ,,,200
        OCTQ    261,12127,10275,1200
ESTFW   SFR     4,ESTS+5                TYPEWRITER SUBROUTINE
        SFR     5,ESTS+6
        SFR     7,ESTS+7
        ATN     3840
        SSR     62
ESTFW2  CAM     14,M13
        JSB     3,ESTQW2
        FIL
        ATN     256
        SSR     62
        CAM     1,4
        LFR     7,ESTS+7
        ATN     13,1,
        LFR     5
        SFR     7,ESTS+7
ESTFW1  ORB     1,1,
        CAM     14,M0
        JSB     3,ESTQW1
        FIL
        CAM     0,M1-8
        JNM     0,ESTFW1
        LFR     7,ESTS+7
        CAM     3,M14-M13
        JPM     3,ESTFW2
        LFR     4,ESTS+5
        LFR     5,ESTS+6
        JLH     M15
        FIL
```

```
TYPKOD  OCTQ   10070,10171,10172,10073,10130,10031,10032,10133
        OCTQ   10174,10075,10040,10177,10002,10103,10144,10045
        OCTQ   10046,10147,10004,10105,10106,10007,10160,10061
        OCTQ   10062,10163,10020,10121,10122,10023,10064,10165
        OCTQ   10166,10067,10124,10025,10026,10127
        FIL
ESTQW3  ATN    512
ESTQW2  ATN    512
ESTQW1  ATN    256
ESTQW   CAM    13
        JPM    14,ESTQW+2
        ATN    M13+4217
        SSR    62
        JNM    14,3,ESTQW+2
        ATN    M13+4152
        SSR    62
        SFR    6,ESTS
        CRN    14,12
        CAM    10
        CAM    11,M15
        CSM    9,4
        JPM    10,ESTQW+6
        ATN    1
        LFR    7,TYPKOD
        CRM    10,10
        ORB    M10
        ATN    M12
        SSR    62
        CJU    9,ESTQW+5
        CAM    15,M11
        LFR    6,ESTS
        JLH    M3
ESTS    BSS    8
        GO     0
```